query: embed.html## #defaultView:Graph
  PREFIX wd: <http://www.wikidata.org/entity/>
  PREFIX wdt: <http://www.wikidata.org/prop/direct/>
  PREFIX gas: <http://www.bigdata.com/rdf/gas#>
  
  select ?citing_work ?cited_workLabel ?cited_work ?citing_workLabel where {
    hint:Query hint:optimizer "None" .
    {
      select ?citing_work ?cited_work where {
        service gas:service {
          gas:program gas:gasClass "com.bigdata.rdf.graph.analytics.BFS" ;
                  gas:in wd:Q27932040 ; 
                  gas:traversalDirection "Undirected" ;
                  gas:out ?citing_work ;
                  gas:out1 ?depth ;
                  gas:out2 ?cited_work1 ;
                  gas:linkType wdt:P2860 ;
                  gas:maxIterations 1;
                  gas:maxVisited 200 .
        }
        ?citing_work wdt:P2860 ?cited_work .
      }
      limit 100
    }
    
    
    	{ 
        ?citing_work (p:P50) ?citing_author_statement . 
        ?citing_author_statement pq:P1545 "1" .
        ?citing_author_statement ps:P50 ?citing_author .
        ?citing_author rdfs:label ?citing_author_name .
        filter(lang(?citing_author_name) = 'en')
      }
      union 
    	{ 
        ?citing_work (p:P2093) ?citing_author_statement . 
        ?citing_author_statement pq:P1545 "1" .
        ?citing_author_statement ps:P2093 ?citing_author_name .
      }
    
      { 
        ?cited_work (p:P50) ?cited_author_statement . 
        ?cited_author_statement pq:P1545 "1" .
        ?cited_author_statement ps:P50 ?cited_author .
        ?cited_author rdfs:label ?cited_author_name .
        filter(lang(?cited_author_name) = 'en')
      }
      union 
    	{ 
        ?cited_work (p:P2093) ?cited_author_statement . 
        ?cited_author_statement pq:P1545 "1" .
        ?cited_author_statement ps:P2093 ?cited_author_name .
      }
  
      ?citing_work wdt:P577 ?citing_date . 
      ?cited_work wdt:P577 ?cited_date . 
      bind(year(?citing_date) as ?citing_year)
      bind(year(?cited_date) as ?cited_year)
      bind(concat(?citing_author_name, ", ", str(?citing_year)) as ?citing_workLabel)
      bind(concat(?cited_author_name, ", ", str(?cited_year)) as ?cited_workLabel)
  }
  